

l'autre d'une embase encartable auto-dénudante que l'on soudera directement sur l'affichage. Le premier connecteur nommé vient s'enficher sur l'embase K2. On remplacera la résistance R1 par un pont de câblage.

Si l'on fait appel à un autre type d'affichage à rétro-éclairage il faudra qu'il s'agisse de rétro-éclairage à LED. Il existe des affichages qui demandent le branchement au +5 V du rétro-éclairage (remplacer R1 par un pont de câblage) et d'autres exigeant une certaine intensité de courant (adapter la valeur de R1 en conséquence). Le cavalier de court-circuit (jumper) **JP1** permet une commande simple du rétro-éclairage de l'affichage.

Il faudra, en fonction du type de commande, en tension ou en courant, adapter la valeur de R1, voire le cas échéant la remplacer par un pont de câblage. Lorsque l'éclairage n'est pas effectué directement par le connecteur mais par l'intermédiaire de 2 lignes distinctes, on pourra couper les conducteurs n°15 et n°16 (15 = BL : connexion de l'anode, 16 = masse : connexion de la cathode). Pour éviter toute erreur de connexion nous vous proposons le brochage dans le **tableau 1**.

Une fois l'affichage relié au bus universel via l'électronique proposée ici, il restera, après avoir mis le PC en marche, à l'initialiser. On devrait, pour le moment, voir une ligne de caractères vierge et une ligne de caractères pleins, ce qui signifie que l'affichage est bien connecté. Si l'on ne voit rien sur l'affichage il peut y avoir un problème de contraste. On jouera sur l'ajustable P1 pour voir si les choses s'arrangent. S'il n'apparaît toujours rien sur l'affichage il faudra vérifier la correction de son branchement.

On pourra, si l'on utilise un affichage à rétro-éclairage, utiliser un multimètre pour vérifier, aux bornes du cavalier JP1 le courant consommé par les LED du ré-

tro-éclairage. Ce courant devrait être compris entre 100 et 150 mA. Si la commande du rétro-éclairage se fait en courant, on adaptera la valeur de R1 de manière à ce qu'il circule le courant convenable à travers les LED. On implantera le cavalier JP1 lorsqu'il faudra alimenter le rétro-éclairage en permanence.

Nous vous proposons, en **tableau 2**, un petit programme montrant comment initialiser l'affichage. Une fois l'initialisation effectuée, on a émission une ligne de test de 80 caractères, processus dont le résultat est l'apparition de caractères à partir de la position 0 de l'affichage.

Tableau 2

```

10 DISP =&H300: REM LES ADRESSES PEUVENT ETRE &H300+(4*X), 0≤X≤7
20 REM DÉFINITION DES ADRESSES DE L'AFFICHAGE
30 DISPREAD=DISP+3
40 DISPWRITE=DISP+2
50 DISPBUSY=DISP+1
60 DISPCTL=DISP+0
70 REM INITIALISATION DE L'AFFICHAGE
80 OUT (DISPCTL),&H30
90 FOR I=0 TO 10: NEXT
100 OUT (DISPCTL),&H30
110 GOSUB 290
120 OUT (DISPCTL),&H30
130 GOSUB 290
140 OUT (DISPCTL),&H38: REM DÉFINIR 2 LIGNES SUR L'AFFICHAGE
150 GOSUB 290
160 OUT (DISPCTL),&HE: REM AFFICHAGE ON, CURSOR ON
170 GOSUB 290
180 OUT (DISPCTL),&H1: REM EFFACER L'AFFICHAGE
190 GOSUB 290
200 OUT (DISPCTL),&H6: REM DÉFINIR MOUVEMENT DU CURSEUR
210 REM -----
220 REM AFFICHER UNE CHAÎNE DE TEST DE 80 CARACTÈRES À L'ÉCRAN
230 REM -----
240 FOR I=32 TO 32+79
250 GOSUB 290
260 OUT (DISPWRITE),(I)
270 NEXT
280 END
290 REM -----
300 REM ----- VOIR SI L'AFFICHAGE EST « BUSY » -----
310 A=INP(DISPBUSY): IF (A AND 128) <>0 THEN 310
320 RETURN

```

Tableau 1

Broche	K2	Affichage
1	GND	Vss
2	+5V	VDD
3	Contraste	V _o
4	A1	RS
5	A0	R/W
6	DISP	E
7	D0	DB0
8	D1	DB1
9	D2	DB2
10	D3	DB3
11	D4	DB4
12	D5	DB5
13	D6	DB6
14	D7	DB7
15	BL	A(node)
16	GND	C(athode)